Introduction to Computable General Equilibrium Model (CGE)

Dhazn Gillig
&
Bruce A. McCarl

Department of Agricultural Economics
Texas A&M University

Course Outline

- **Overview of CGE**
- **An Introduction to the Structure of CGE**
- **An Introduction to GAMS**
- **Casting CGE models into GAMS**
- **Data for CGE Models & Calibration**
- **Incorporating a trade & a basic CGE application**
- **Evaluating impacts of policy changes and casting nested functions & a trade in GAMS**
- Mixed Complementary Problems (MCP)

This Week's Road Map

- **Discussions on technical matters on modeling that have not been discussed in the class.**
  - **Deriving welfare impacts of policy changes**
  - **Building nested functions in GAMS**
  - **Extending a closed economy to a small open economy in GAMS**
- **CGE application of Manne & Richels**

Evaluating impacts of policy changes

Benchmark Equilibrium

Shocks

Counterfactual Equilibrium

Prices and quantities => household incomes, consumer utility levels, etc.

Prices and quantities => household incomes, consumer utility levels, etc.

Who has gained
OR has lost
from a policy change?

Which tool should be used to make a comparison between equilibriums?

Evaluating impacts of policy changes

There are several approaches to evaluate impacts of policy changes; however, most of the CGE literature on the effects of policy changes focus on welfare measures:

1. Compensating Variation (CV)
2. Equivalent Variation (EV)

Welfare measures – CV

Compensating Variation:

How much money is necessary to compensate someone for price changes?

$CV = E(U^1,P^1) - E(U^0,P^1)$ => $CV = [(U^1 - U^0)/ U^1] * Y^1$

STEPS to calculate CV:

1. **Calculate a new utility level**

$$U = \sum_j \left[(\alpha_j)^{1/\sigma} (X_j)^{(\sigma-1)/\sigma} \right]^{\sigma/(\sigma-1)}$$

2. **Calculate a new income level**

$$Y^1 = W_L^1 \overline{L}_h + W_K^1 \overline{K}_h + Transfers$$

3. **Calculate a utility difference**

Welfare measures –EV

Equivalent Variation:

How much money is a particular change equivalent to?

EV = E(U^1,P^0) - E(U^0,P^0) => EV = [(U^1 - U^0)/ U^0] * Y^0

STEPS to calculate EV:

1. **Calculate a new utility level**

$$U = \sum_j \left[(\alpha_j)^{1/\sigma} (X_j)^{(\sigma-1)/\sigma} \right]^{\sigma/(\sigma-1)}$$

2. **Calculate a utility difference**

Price & quantity measures

An alternative to evaluating impacts of policy changes is **price and quantity measures**. These measures involve developing price and quantity indexes which can be used to describe how large adjustments are between a base scenario and an alternative scenario.

The simplest price and quantity measures are:

1. Laspeyres price index
2. Laspeyres quantity index
3. Paasche price index
4. Paasche quantity index

Price & quantity measures

The Laspeyres price index: $L^P = \sum_j P_j^1 X_j^0 / \sum_j P_j^0 X_j^0$

shows the ratio between the aggregate value of all commodities at prices in the new equilibrium but quantities in the old equilibrium and the aggregate value of all commodities at the old equilibrium prices and quantities.

The Laspeyres quantity index: $L^Q = \sum_j P_j^0 X_j^1 / \sum_j P_j^0 X_j^0$

shows the ratio between the aggregate value of all commodities at quantity in the new equilibrium but prices in the old equilibrium and the aggregate value of all commodities at the old equilibrium prices and quantities.

Note that: This price index is similar to EV where we compare the aggregate value of all goods with the old equilibrium value of all goods.

P's are prices and X's are quantity. Subscripts '1' and '0' refer to a new and old equilibriums, respectively.

Price & quantity measures

The Paasche price index: $P^P = \sum_j P_j^1 X_j^1 / \sum_j P_j^0 X_j^1$

shows the ratio between the aggregate value of all commodities considered at prices and quantities in the new equilibrium and the aggregate value of all commodities at the new equilibrium quantities but prices in the old equilibrium.

The Paasche quantity index: $P^Q = \sum_j P_j^1 X_j^1 / \sum_j P_j^1 X_j^0$

shows the ratio between the aggregate value of all commodities considered at prices and quantities in the new equilibrium and the aggregate value of all commodities at prices in the new equilibrium but quantities in the old equilibrium.

Note that: This price index is similar to CV where we compare the aggregate value of all goods with the new equilibrium value of all goods.

Building nested functions in GAMS

Suppose we want to put nested functions in GAMS.

Recall: With the use of nested functions, a system allows substitution in the model

Assumptions:

1. Leontief technology using INT and VA **at a top level**
2. Cobb Douglas technology using L and K **at a bottom level**
3. Non-nested CES utility function

Building nested functions in GAMS – Modifications on variables

```
VARIABLE
   TaxRevenue                        Total government tax revenues  ;
POSITIVE VARIABLE
   FactorPrice(AllSets)              Factor price
   FactorQuan(AllSets,AllSets1)      Factor use by a producing sector
   ComPrice(AllSets)                 Commodity price
   DemCommod(AllSets,AllSets1)       Commodity demand by households
   Production(AllSets)               Production quantity levels
   HHIncome(AllSets)                 Household income
   QIntA(AllSets,AllSets1)           Intermedite inputs quantity
   PIntA(AllSets,AllSets1)           Intermedite inputs price
   QValAdd(AllSets)                  Value-added quantity
   PValAdd(AllSets)                  Value-added price
    ;
```

Building nested functions in GAMS – Modifications on equations

```
EQUATIONS
    FactorMkt(AllSets)                 Factor market balances
    FactorDem(AllSets,AllSets1)        Factor demand by a sector
    CommodMkt(AllSets)                 Commodity market balance
    CommodDem(AllSets,AllSets1)        Commodity Demand by Households
    Profit(AllSets)                    Zero profit condition
    Income(AllSets)                    Household budget constraint
    GovBal                             Government budget constraint

    QintAEq(AllSets,AllSets1)          Intermedite inputs quantity equation
    PIntAEq(AllSets,AllSets1)          Intermedite price equation
    CDQVAEq(AllSets)                   Value-added quantity equation
    PVAEq(AllSets)                     Value-added inputs price equation

      ;
```

Building nested functions in GAMS – Modifications on VA

Steps:

1. **Define quantities of value-added (QVA)**
2. **Define prices of VA**
3. **Define factor demands used in producing VA**

$$QVA_j = A_j K_j^{\alpha} L_j^{1-\alpha} \longrightarrow QVA_j = A_j \prod_f F_{fj}^{\alpha_{fj}}$$

```
CDQVAEq(Sector)..
 QValAdd(Sector)
 =E=  BigA(Sector)
 *(PROD(Factor,FactorQuan(Factor,Sector)**AlphaCD(Factor,Sector) ) ) ;
```

Building nested functions in GAMS – Modifications on VA

Prices of VA are derived from the relationship between revenues and costs of producing final goods where **revenue (PxQ) is exhausted by payments for VA and INT inputs.**

```
PVAEq(Sector)..
  ComPrice(Sector)* Production(Sector)
   =G=
  PValAdd(Sector)*QValAdd(Sector)
+ SUM(Activity,PIntA(Activity,Sector) * QIntA(Activity,Sector)
     $YesQIntA(Activity,Sector)  )      ;
```

```
PARAMETER  YesQIntA(AllSets,AllSets1)     Yes there are intermediate inputs;

YesQIntA(Activity,Sector)            = NO  ;
YesQIntA("Food","NonFood")           = YES ;
YesQIntA("NonFood","Food")           = YES ;
```

Building nested functions in GAMS – Modifications on VA

Factor demand : $F_{fj} = \frac{QVA_j}{A_j}\left(\frac{\alpha_{fj}\, W_{f'}(1+t_{f'})}{\alpha_{f'j}\, W_f(1+t_f)}\right)^{\alpha_{f'j}}$

```
FactorDem(Factor,Sector)..
 FactorQuan(Factor,Sector)
  =E=
 (QValAdd(Sector) / BigA(Sector))
  * SUM(MapFact(Factor,Factor1),
            (  ( AlphaCD(Factor,Sector)
                   *(FactorPrice(Factor1)*(1 + TaxFactor(Factor1))) )
                     / (AlphaCD(Factor1,Sector)
                          * (FactorPrice(Factor)*(1 +TaxFactor(Factor))) )
            )**(AlphaCD(Factor1,Sector))
         ) ;
```

Building nested functions in GAMS – Modifications on INT

Steps:

1. Define quantities of intermediate input (QINT)
2. Define prices of INT

$$QINTA_{j1j} = a_{j1j} Q_j$$ where $a_{j1,j}$ is I-O coefficients, using j1 to produce j

```
QIntAEq(Sector,Activity)..
 QIntA(Sector,Activity)
   =E=
 AlphaLeon(Sector,Activity) * Production(Activity)      ;
```

```
AlphaLeon(Sector,Activity)
= SAM(Sector,Activity) /SAM("Total",Activity) ;
```

Building nested functions in GAMS – Modifications on INT

Intermediate input prices depend on commodity prices and intermediate input coefficient representing the quantity of input per unit of INT input (not output).

```
PIntAEq(Activity,Sector)..
 PIntA(Activity,Sector)
  =E=
 AlphaLeonc(Activity,Sector) * ComPrice(Sector)    ;
```

```
AlphaLeonc(Activity,Sector)
 = SAM(Activity,Sector) /SUM(Sector1,SAM(Activity,Sector1)) ;
```

Building nested functions in GAMS – Complementarity

```
MODEL CGEModel

        / FactorMkt.FactorPrice
          FactorDem.FactorQuan
          Commoddem.DemCommod
          CommodMkt.ComPrice
          Profit.Production
          Income.HHincome
          Govbal.TaxRevenue
          QintAEq.QintA
           PIntAEq.PIntA
           CDQVAEq.QValAdd
           PVAEq.PValAdd

          /;
```

Incorporating a small open economy in GAMS

Now, we are going to see how to incorporate a small open economy in GAMS.

Assumptions:

1. World prices and an exchange rate are exogenous.
2. Elasticity of substitution in household CES utility function = 0.7
3. Free of taxes at the benchmark equilibrium
4. Let an exchange rate be numeriare.
5. **Counterfactual equilibrium**

   If an export tax is imposed, then 100% of tax revenues is used to **purchase the government goods/services.**

Incorporating a small open economy in GAMS – Modifications on variables

```
VARIABLE
   TaxRevenue                          Total government tax revenues  ;
POSITIVE VARIABLE
   FactorPrice(AllSets)                Factor price
   FactorQuan(AllSets,AllSets1)        Factor use by a producing sector
   ComPrice(AllSets)                   Commodity price
   DemCommod(AllSets,AllSets1)         Commodity demand by households
   Production(AllSets)                 Production quantity levels
   HHIncome(AllSets)                   Household income
   QIntA(AllSets,AllSets1)             Intermedite inputs quantity
   PIntA(AllSets,AllSets1)             Intermedite inputs price
   QValAdd(AllSets)                    Value-added quantity
   PValAdd(AllSets)                    Value-added price
   PExp(AllSets)                       US Export price paid by ROW but received
   PImp(AllSets)                       US Import price paid by US consumers but
   QExp(AllSets)                       US Export quantity
   QImp(AllSets)                       US Import quantity
   PWExp(AllSets)                      ROW or world export price
   PWImp(AllSets)                      ROW or world import price
   Exchange                            Exchange rate
   ;
```

Incorporating a small open economy in GAMS – Modifications on equations

```
EQUATIONS
    FactorMkt(AllSets)                  Factor market balances
    FactorDem(AllSets,AllSets1)         Factor demand by a sector
    CommodMkt(AllSets)                  Commodity market balance
    CommodDem(AllSets,AllSets1)         Commodity Demand by Households
    Profit(AllSets)                     Zero profit condition
    Income(AllSets)                     Household budget constraint
    GovBal                              Government budget constraint

    QintAEq(AllSets,AllSets1)          Intermedite inputs quantity equation
    PIntAEq(AllSets,AllSets1)          Intermedite price equation
    CDQVAEq(AllSets)                   Value-added quantity equation
    PVAEq(AllSets)                     Value-added inputs price equation

    PExpBal(AllSets)                    US Export price
    PImpBal(AllSets)                    US Import price
    QExpBal(AllSets)                    US Export demand equation
    QImpBal(AllSets)                    US Import supply equation
    PUSExpPBal(AllSets)                 US domestic export price relationship
    PUSImpPBal(AllSets)                 US domestic import price relationship
    TradeBal                            Trade balance
      ;
```

--

1. The commodity market balance

$$Q_j \geq \sum_h X_{hj} + \sum_{j1} a_{j,j1} QIntA_{j\,j1} + s_j R / P_j + QExp_j - QImp_j$$

```
CommodMkt(Sector)..
    Production(Sector)
     =G=
    sum(Households,DemCommod(Households,Sector))
+ sum(OtherSector$QintA0(Sector,OtherSector),
      QIntA(Sector,OtherSector))
+ GovTaxShare(Sector)*(TaxRevenue/ComPrice(Sector))
+ QExp(Sector)$ExTrade(Sector)
-  QImp(Sector)$ImTrade(Sector)     ;
```

2. The government tax revenue balance

$$R \le \sum_h \left(t_h \sum_f \bar{F}_{fh} W_f \right) + \sum_{fj} t_{fj} W_f F_{fj} + \sum_{j \in ex} tExp_j QExp_j P_j + \sum_{j \in im} tImp_j QImp_j P_j$$

```
GovBal..
   TaxRevenue
   =L=
    SUM(Households,
          Incometax(Households)
           * SUM(Factor,Endowment(Factor,HouseHolds)* FactorPrice(Factor))    )
  + SUM((Factor,Sector),TaxFactor(Factor,Sector)*FactorPrice(Factor)
          *FactorQuan(Factor,Sector) )
  + SUM(Sector,(TaxExp(Sector)
        *QExp(Sector)*ComPrice(Sector))$ExTrade(Sector))
  + SUM(Sector,(TaxImp(Sector)
        *QImp(Sector)*ComPrice(Sector))$ImTrade(Sector))            ;
```

3. The domestic trade price equations

$$(PWExp_j \times Exchange) - TransCost_j \geq PExp_j \qquad \forall j \in ex$$

```
PExpBal(sector)$ExTrade(Sector)..
 (PWExp(Sector)*Exchange) -Transcost(Sector)
 =G=     PExp(Sector)                              ;
```

$$(PWImp_j \times Exchange) + TransCost_j \geq PImp_j \qquad \forall j \in im$$

```
PImpBal(sector)$ImTrade(Sector)..
 (PWImp(Sector)*Exchange) +Transcost(Sector)
 =G=     PImp(Sector)                              ;
```

4. The domestic trade quantity equations

$$QExp_j = f(PWExp_j, \varepsilon) = a_j * PWExp_j^{\varepsilon_j} \quad \forall j \in ex$$

```
QExpBal(sector)$ExTrade(Sector)..
 QExp(Sector)   =E=
 ExpDem("cons",Sector) *(PWExp(Sector)**ExpDem("slope",Sector))  ;
```

$$QImp_j = f(PWImp_j, \mu) = b_j * PWImp_j^{\mu_j} \quad \forall j \in im$$

```
QImpBal(sector)$ImTrade(Sector)..
 QImp(Sector)   =E=
 ImpDem("cons",Sector) *(PWImp(Sector)**ImpDem("slope",Sector)) ;
```

5. The domestic & trade price relationship

$$PExp_j = (1 - tExp_j)\, P_j \qquad \forall\, j \in ex$$

```
PUSExpPBal(Sector)$ExTrade(Sector)..
  PExp(Sector)   =E=
  (1-TaxExp(Sector)) * ComPrice(Sector)   ;
```

$$PImp_j = (1 + tImp_j)\, P_j \qquad \forall\, j \in im$$

```
PUSImpPBal(Sector)$ImTrade(Sector)..
  PImp(Sector)   =E=
  (1+TaxImp(Sector)) * ComPrice(Sector)   ;
```

6. The zero trade balance

$$\sum_{j \in im} PWImp_j QImp_j \leq \sum_{j \in ex} PWExp_j QExp_j$$

```
TradeBal..
  SUM(Sector$ImTrade(Sector),PWImp(Sector)*QImp(Sector))
   =L=
  SUM(Sector$ExTrade(Sector),PWExp(Sector)*QExp(sector))
        ;
```

Incorporating a small open economy in GAMS – Complementarity

```
MODEL CGEModel

      / FactorMkt.FactorPrice
        FactorDem.FactorQuan
        Commoddem.DemCommod
        CommodMkt.ComPrice
        Profit.Production
        Income.HHincome
        Govbal.TaxRevenue

        QintAEq.QintA
        PVAEq.PValAdd
        PIntAEq.PIntA
        CDQVAEq.QValAdd

        Qexpbal.Qexp
        Qimpbal.Qimp
        PUSexpPbal.Pexp
        PUSimpPbal.Pimp
        Pexpbal.Pwexp
        Pimpbal.Pwimp
        Tradebal.Exchange          /;
```

Incorporating a small open economy in GAMS – Results

```
----    1112 PARAMETER Compare  Comparative analysis

                                   NoTax    ExpTax20%

ExpPrice      .Food                1.000        1.072
ImPrice       .NonFood             1.000        1.725
USExPrice     .Food                1.000        1.340
USIMPrice     .NonFood             1.000        1.725
```

$$PExp_j = (1 - tExp_j) P_j$$

distortion in prices

Incorporating a small open economy in GAMS – Results

```
----    1112 PARAMETER Compare  Comparative analysis
```

		NoTax	ExpTax20%
HH demand	.Food	1000.000	656.847
HH demand	.NonFood	900.000	591.162
Gov demand	.Food		30.000
Gov demand	.NonFood		23.301
Int demand	.Food	100.000	57.965
Int demand	.NonFood	200.000	149.259
Domestic Demand	.Food	1100.000	744.812
Domestic Demand	.NonFood	1100.000	763.722
Export Demand	.Food	300.000	300.000
Demand	.Total	2500.000	1808.534
Domestic Supply	.Food	1400.000	1044.812
Domestic Supply	.NonFood	800.000	463.722
Import Supply	.NonFood	300.000	300.000
Supply	.Total	2500.000	1808.534

$\rightarrow s_j R / P_j$

How do you explain these results?

Key Elements – MERGE

MERGE:

- Model for Evaluating Regional and Global Effects
- Stanford University & Electric Power Research Institute
- Multi-sectors CGE model
- 9 regions
- The energy sector
- International trade => carbon emission rights
- Non-energy & energy inputs
- Changes in the cost of energy => production
- Sinks and non-CO2 gas

More on MERGE see http://www.stanford.edu/group/MERGE/

Wrap Up

- Evaluating results from CGE models
- Incorporating nested functions and a trade relationship in GAMS

Next:

- MCP
- MacCracken, C. N., J. A. Edmonds, S. H. Kim, and R. D. Sands. "The Economics of the Kyoto Protocol," in *The Costs of the Kyoto Protocol: A Multi-Model Evaluation*, John Weyant (ed.), special issue of *The Energy Journal*, 1999.
- Incorporating environmental aspects (e.g. Ghg emissions) in the CGE model and in GAMS

Reference:

Shoven, J. B. and J. Whalley. "Applying general equilibrium." ***Surveys of Economic Literature***, Chapter 5, 1998.

Manne A. S. and R. G. Richels. "An alternative approach to establishing trade-offs among greenhouse gases." ***Nature*** 410, 675-677 (2001).